Protokollfunktion für Ihre Datenbanken Teil 1

Bei umfangreicheren Datenbank-Anwendungen, die beispielsweise in verschiedenen Abteilungen oder bei mehreren Kunden eingesetzt werden, treten immer wieder Probleme auf, für deren Lösung Sie detaillierte Informationen zum System und zu ausgeführten Arbeitsschritten benötigen. Praktisch ist in solchen Fällen eine Protokolldatei ("Log File"), in das während der Arbeit mit der Datenbank verschiedene, für Sie wichtige Informationen geschrieben werden. Bei Problemen lassen Sie sich die Protokolldatei zusenden oder greifen per Netzwerk darauf zu und können so die Ursache schneller finden. Wird zum Beispiel die aktuelle Access-Version festgehalten, können Sie sofort erkennen, ob das aktuellste Service Pack installiert ist. Angaben zum Hauptspeicher helfen beispielsweise, wenn die Ausführung umfangreicher Abfragen oder das Öffnen größerer Formulare und Berichte Probleme erzeugt. In diesem zweiteiligen Tipp der Woche stellen wir Ihnen eine universell einsetzbare Lösung vor, mit deren Hilfe Sie eine solche Protokolldatei flexibel in Ihren Datenbanken nutzen können.

Im ersten Teil geht es zunächst um eine globale Prozedur "WriteLOG", die Sie an entsprechenden Stellen aufrufen, um Informationen in die Protokolldatei schreiben zu lassen. Der zweite Teil stellt verschiedene Funktionen vor, mit deren Hilfe Sie nützliche Informationen wie Hauptspeicherausbau, Bildschirmauflösung, System, Netzwerkanmeldung, Computername u.s.w. in Kombination mit "WriteLOG" festhalten können.

Der Prozedur "WriteLOG" wird als Parameter "strInfo" lediglich die zu protokollierende Information übergeben. Den Namen und Speicherort der Protokolldatei ermittelt die Prozedur automatisch: Speicherort ist das Verzeichnis, in dem die Datenbankdatei gespeichert ist. Der Name setzt sich aus dem Namen der Datenbank, dem Zusatz "_LOG" und der Erweiterung TXT zusammen. Das Protokoll kann also mit jedem Texteditor geöffnet werden. Der zu protokollierenden Information stellt die Prozedur außerdem automatisch Datum und Zeit voran. Ein Aufruf wie

WriteLOG "Access " & SysCmd(acSysCmdAccessVer) & _
                 " gestartet..."

würde dann zum Beispiel zu einem Eintrag wie

12.08.2005 13:47 - Access 11.0 gestartet...

führen und Ihnen anzeigen, dass der Mitarbeiter oder Kunde Access 2003 einsetzt.

Für den grundsätzlichen Einsatz der Lösung benötigen Sie die Prozedur "WriteLOG" sowie zwei Hilfsfunktionen "GetDatabasePath()" und "GetDatabaseName()". Legen Sie zunächst ein neues Modul "modWriteLOG" an und geben Sie diese Prozedur bzw. diese Funktionen dort ein:

Function GetDatabaseName() As String
  Dim strName As String
  Dim L As Integer

  strName = CurrentDb.Name
  L = Len(strName)
  While Mid$(strName, L, 1) <> "\" And L > 0
    L = L - 1
  Wend
  If L = 1 Then Exit Function
  strName = Mid$(strName, L + 1)

  L = Len(strName)
  While Mid$(strName, L, 1) <> "." And L > 0
    L = L - 1
  Wend
  If L = 1 Then Exit Function
  strName = Left$(strName, L - 1)
  GetDatabaseName = strName

End Function

Function GetDatabasePath() As String
    Dim L As Integer
    Dim strPath As String

    strPath = CurrentDb.Name
    L = Len(strPath)
    While Mid$(strPath, L, 1) <> "\" And L > 0
        L = L - 1
    Wend
    If L > 1 Then
        strPath = Left$(strPath, L)
    Else
        strPath = ""
    End If
    GetDatabasePath = strPath

End Function

Sub WriteLOG(strInfo As String)
  Dim strPath As String, strFName As String
  Dim FNum As Integer

  On Error Resume Next
  strPath = GetDatabasePath()
  strFName = strPath & GetDatabaseName() & "_LOG.txt"
  FNum = FreeFile
  Open strFName For Append As #FNum
  Print #FNum, Format$(Now, "dd.mm.yyyy - hh:mm:ss") & _
               " - " + strInfo
  Close #FNum
  DoEvents
  Err.Clear

End Sub

Die Prozedur "WriteLOG" setzt zunächst mit Hilfe von "GetDatabasePath()" und "GetDatabaseName()" gemäß Name und Speicherort der aktuellen Datenbank den Dateinamen und Speicherort der Protokolldatei zusammen. Mit diesem Namen wird eine Textdatei im Modus "Append" (anhängen) geöffnet und die zu protokollierende Information mit Datums-/Zeit-Zusatz hineingeschrieben. Anschließend wird die Protokolldatei wieder geschlossen.

Wenn Sie nun den Direktbereich öffnen und dort

WriteLOG "Dies ist ein Test..."

eingeben, haben Sie die Protokollfunktion ein erstes Mal erfolgreich genutzt. Im Verzeichnis der Datenbank finden Sie die TXT-Datei mit diesem Protokolleintrag.

Soviel zunächst für diese Ausgabe. In der nächsten Woche stellen wir Ihnen verschiedene Funktionen für den Abruf weitergehender Access-, Netzwerk- und Systeminfos vor und zeigen, wie diese in Verbindung mit "WriteLOG" in einer Datenbank an verschiedenen Stellen (AutoExec, Form_Open, After_Update u.s.w.) eingesetzt werden können.
